Jacques Garrigue, 2010 年 11 月 17 日

再帰的アルゴリズム2

9　再帰データ型 (続き)

木構造

再帰データ型の定義は線形でなくてもいい．よく知られている例はファイルシステムなどに使われる木構造である．

```
type fs =
    File of string                                          (* ファイルの中身 *)
  | Directory of (string * fs) list ;;
                                  (* 名前とファイルまたはフォルダーの連想リスト *)

let desktop = Directory ["test.ml", File "let x = 1;;"] ;;
val desktop : fs = ...
let home =
  Directory ["Desktop", desktop; ".emacs", File "(require 'caml-font)"] ;;
val home : fs = ...

let rec lookup path fs =
  match path, fs with                              (* path と fs を同時にマッチする *)
  | name :: rem, Directory dir ->                    (* ファイルをフォルダーで探す *)
      lookup rem (List.assoc name dir)                  (* 連想リストをキーで探す *)
  | [], _ -> fs                            (* ファイルまたはフォルダーが見つかった *)
  | _ -> failwith "lookup"                              (* ファイルの中には入れない *)
;;
val lookup : string list -> fs -> fs = <fun>

lookup ["Desktop"; "test.ml"] home;;
- : fs = File "let x = 1;;"
```

練習問題 9.3 特定の場所にファイルやフォルダーを挿入する関数を書きなさい．

```
val add : string list -> fs -> fs -> fs
```

抽象構文と評価

木構造のもう一つの応用は抽象構文の表現である．
以下のような言語を扱う処理系を作りたい．

$$\text{式} ::= \text{実数} \mid \text{変数} \mid \text{式} + \text{式} \mid \text{式} \times \text{式} \mid \sin(\text{式}) \mid \cos(\text{式}) \mid (\text{式})$$

式の例

$$5 \times 2 + 3 \qquad (3 + y) \times 12$$

抽象構文を表す型をまず定義する．

```
type expr =
    X                                              (* 変数は X だけ *)
  | Cst of float                                   (* 定数には実数を使う *)
  | Add of expr * expr
  | Mul of expr * expr
  | CMul of float * expr                           (* 定数の掛け算 *)
  | Sin of expr
  | Cos of expr

let rec eval e x =                                 (* 評価は再帰関数 *)
  match e with
    X     -> x
  | Cst a -> a
  | Add (e1, e2) -> eval e1 x +. eval e2 x
  | Mul (e1, e2) -> eval e1 x *. eval e2 x
  | CMul (a, e1) -> a *. eval e1 x
  | Sin e1 -> sin (eval e1 x)
  | Cos e1 -> cos (eval e1 x)

# eval (Mul (Sin X, Sin X)) 1.;;
- : float = 0.708073418273571176
```

こんな式に対して，微分が簡単に定義できる．

```
let derive_op1 = function                          (* 引数に対する微分 *)
    CMul(a, x) -> Cst a
  | Sin x -> Cos x
  | Cos x -> CMul(-1., Sin x)
  | _ -> failwith "derive_op1"
val derive_op1 : expr -> expr

let get_arg = function                             (* 1引数の構文の引数を返す *)
    CMul (_, e1) | Sin e1 | Cos e1 -> e1
  | _ -> failwith "get_arg"                        (* それ以外の構文でエラー *)

let rec derive = function
    X -> Cst 1.
  | Cst _ -> Cst 0.
  | Add (e1, e2) -> Add(derive e1, derive e2)
  | Mul (e1, e2) -> Add (Mul(derive e1, e2), Mul(e1, derive e2))
  | e -> Mul (derive (get_arg e), derive_op1 e)

# derive (Mul (Sin X, Cos X));;
- : expr =
Add (Mul (Mul (Cst 1., Cos X), Cos X),
 Mul (Sin X, Mul (Cst 1., CMul (-1., Sin X))))
```

ここで **function** というキーワードを使っているが，以下の構文の略になる．

```
fun e -> match e in
```

短かい上に引数の名前を考えなくていいから便利だ．

練習問題 9.4　*1.* 様々な式を微分し，正しさを確かめよ．問題 *4.1.2* の derive とも比較せよ．

2. 新しい基本関数を追加せよ．

式の単純化

微分を正しく定義できたものの，式をそのまま出すと冗長な部分が多い．やはり単純しなければならない．

```
# derive (Sin (Add(X, Cst 1.)));;
- : expr = Mul (Add (Cst 1., Cst 0.), Cos (Add (X, Cst 1.)))
```

まず考えられるのは，変数を含まない部分を完全に計算してしまうことだ．

```
let map f e =                                              (* map も要る *)
  match e with
    X | Cst _ -> e
  | Add (e1, e2) -> Add (f e1, f e2)
  | Mul (e1, e2) -> Mul (f e1, f e2)
  | CMul (a, e1) -> CMul (a, f e1)
  | Sin e1 -> Sin (f e1)
  | Cos e1 -> Cos (f e1)

let rec const = function                                   (* X を含むかどうか *)
    X -> false
  | Cst _ -> true
  | Add (e1, e2) | Mul (e1, e2) -> const e1 && const e2
  | e -> const (get_arg e)

let rec simpl e =
  if const e then Cst (eval e 0.) else map simpl e ;;

# simpl (derive (Sin (Add(X, Cst 1.))));;
- : expr = Mul (Cst 1., Cos (Add (X, Cst 1.)))
```

これで余計な部分が少し減ったが，まだまだ残っている．
次の考え方は，単純化に使える方程式を利用する．例えば，上の例では，

$$\texttt{Mul}(\texttt{Cst 1.}, \texttt{E}) = \texttt{E}$$

は知られている．

```
exception Nomatch

let simpl1 = function
  | Add(Cst 0., e1) | Add(e1, Cst 0.)           -> e1
  | Mul(Cst a, e1) | Mul(e1, Cst a)             -> CMul(a, e1)
  | Mul(CMul(a, e1), e2) | Mul(e1, CMul(a, e2)) -> CMul(a, Mul(e1, e2))
  | CMul(1., e1) -> e1
  | CMul(0., e1) -> Cst 0.
  | CMul(a, CMul(b, e1)) -> CMul(a*.b, e1)
  | _ -> raise Nomatch

let rec simpl e =
  if const e then Cst (eval e 0.) else
  let e = map simpl e in
  try simpl (simpl1 e) with Nomatch -> e ;;

# simpl (derive (Sin (Add(X, Cst 1.))));;
- : expr = Cos (Add (X, Cst 1.))
```

多項式と正規化

こんな単純化方法では限界がある．例えば，

```
# simpl (derive (Mul (Sin X, Cos X)));;
- : expr = Add (Mul (Cos X, Cos X), CMul (-1., Mul (Sin X, Sin X)))
```

のような結果を多項式として表示したいが，上の構文には多項式を表す方法がない．
単純化を越えて，式を正規化するために，Add と Mul を多項式に変えるといい．
話をはっきりさせるために，もっと制限された式に戻す．

```
type expr0 =
    PowX of float                                          (* X の a 乗 *)
  | Cst of float
  | Add of expr0 * expr0
  | Mul of expr0 * expr0
```

これだったら，変数 X の多項式を次数と係数のリストと見做せばよい．

```
type poly = (float * float) list
let eval_poly p x =
  List.fold_left (fun r (expn, coeff) -> r +. coeff *. (x ** expn)) 0. p
```

もっとも重要な操作は多項式の和だが，次数の順序で並べられていると仮定すると効率よくできる．

```
let rec add_poly p1 p2 =
  match p1, p2 with
    [], _ -> p2
  | _, [] -> p1
  | (x,a)::p1', (y,b)::p2' ->
      if x = y then
        if a +. b = 0. then add_poly p1' p2'
        else (x,a+.b)::add_poly p1' p2'
      else if x < y then (x,a)::add_poly p1' p2
      else (y,b)::add_poly p1 p2'
```

練習問題 9.5 *1.* この add_poly を使って，多項式の掛け算 mul_poly を定義せよ．

2. add_poly と mul_poly を使って，expr0 を poly に変換する関数 poly_of_expr0 : expr0 -> poly を定義せよ．

変数の数を増やすと，上の定義では足りない．

```
type expr1 =
    Pow of string * float
  | ...
type poly = ((string * float) list * float) list
```

数式で書くと

$$P = \Sigma_i(\Pi_j X_{ij}^{a_{ij}}) \cdot b_i$$

このとき，add_poly の定義は変えなくていい．mul_poly では (string * float) list の掛け算を行わなければならないが，実は add_poly がそこでも使える．

練習問題 9.6 多変数多項式における mul_poly および poly_of_expr1 を定義せよ．